
Which Excel? XLOOKUP, UNIQUE and FILTER need Excel 2021 or later, Microsoft 365, or the free Excel for the web. On older versions, INDEX with MATCH (number 11) does XLOOKUP's job. LOOK UP SUMMARISE DECIDE LISTS TIDY

EXCEL FUNCTIONS
The ones we reach for most when cleaning, summarising and reporting data. Every example below runs on the

small table on the right, so you can type it in and check the answer.

S A M P L E  T A B L E  U S E D  I N  E V E R Y  E X A M P L E

A B C D E

1 Facility State Date Tested Positive

2 Kubwa PHC FCT 15/03/2026 120 48

3 Gwale MCH Kano 22/03/2026 95 52

4 Nyanya GH FCT 04/04/2026 0 0

5 Fagge PHC Kano 11/04/2026 140 61

6 Ikeja GH Lagos 18/04/2026 110 23

01 L O O K  U P

XLOOKUP
Finds a value in one column and brings

back the matching value from another.

XLOOKUP(lookup_value, lookup_array,

return_array, [if_not_found])

fx =XLOOKUP("Fagge PHC",A:A,D:D)

RETURNS 140

TIP Add "Not found" as the fourth argument, so

a name that is missing shows that instead of #N/A.

02 S U M M A R I S E

SUMIFS
Adds up the numbers in rows that

meet one or more conditions.

SUMIFS(sum_range, criteria_range1,

criteria1, ...)

fx =SUMIFS(D:D,B:B,"Kano")

RETURNS 235

TIP The column you add up comes first.

Each condition after it is a pair: the range,

then what to look for.

03 S U M M A R I S E

COUNTIFS
Counts the rows that meet every

condition you give it.

COUNTIFS(criteria_range1, criteria1,

...)

fx
=COUNTIFS(B:B,"FCT",D:D,

">0")

RETURNS 1 only Kubwa
PHC

TIP Comparisons go inside the quotes:

">0", "<>Kano", ">=100".

04 S U M M A R I S E

AVERAGEIFS
Averages the numbers in rows that

meet your conditions.

AVERAGEIFS(average_range,

criteria_range1, criteria1, ...)

fx =AVERAGEIFS(D:D,B:B,"FCT")

RETURNS 60

TIP A zero is averaged in; a blank cell is

skipped. Nyanya GH's 0 takes FCT from

120 down to 60.

05 D E C I D E

IF
Gives one result when a test is true

and another when it is false.

IF(logical_test, value_if_true,

value_if_false)

fx
=IF(D2>=100,"High",

"Normal")

RETURNS High

TIP Text results need quotes; numbers and

cell references do not. More than two

outcomes? See IFS, number 15.

06 D E C I D E

IFERROR
Swaps an error for a value you choose,
such as 0 or a blank.

IFERROR(value, value_if_error)

fx =IFERROR(E4/D4,0)

RETURNS 0 not #DIV/0!

TIP Wrap a formula only once it works. Added too

early, it hides real mistakes along with the

expected ones.

07 L I S T S

UNIQUE
Lists each different value once, in the
order it first appears.

UNIQUE(array, [by_col],

[exactly_once])

fx =UNIQUE(B2:B6)

RETURNS FCT

Kano

Lagos

spills down

TIP Keep the cells below it empty.

Anything in the way gives #SPILL!.

08 L I S T S

FILTER
Brings back only the rows that meet a
condition.

FILTER(array, include, [if_empty])

fx =FILTER(A2:A6,B2:B6="Kano")

RETURNS Gwale MCH

Fagge PHC

spills down

TIP Give it a third argument such as

"None". Without one, no match means

#CALC!.

09 T I D Y

TRIM
Removes spaces at the start and end,
and leaves one space between

words.

TRIM(text)

fx =TRIM(A8)

where A8 holds ··Gwale···MCH·

RETURNS Gwale MCH

TIP Trim names before a lookup. "Kano"

with a space after it will never match

"Kano".

10 T I D Y

TEXT
Shows a date or number in the format
you choose, for labels and reports.

TEXT(value, format_text)

fx =TEXT(C2,"mmm yyyy")

RETURNS Mar 2026

TIP The result is text, not a date. Sort and

calculate on the original Date column.

Which Excel? IFS, MAXIFS and TEXTJOIN need Excel 2019 or later; SORTBY and UNIQUE need Excel 2021, Microsoft 365 or Excel for the web. LOOK UP SUMMARISE DECIDE LISTS TIDY DATES

EXCEL FUNCTIONS
Ten more: lookups that work in any version of Excel, cleaner text, sorted lists and reporting dates. The same

sample table, so you can keep checking your answers.

S A M P L E  T A B L E  U S E D  I N  E V E R Y  E X A M P L E

A B C D E

1 Facility State Date Tested Positive

2 Kubwa PHC FCT 15/03/2026 120 48

3 Gwale MCH Kano 22/03/2026 95 52

4 Nyanya GH FCT 04/04/2026 0 0

5 Fagge PHC Kano 11/04/2026 140 61

6 Ikeja GH Lagos 18/04/2026 110 23

1 1 L O O K  U P

INDEX + MATCH
The lookup that works in every version

of Excel. MATCH finds the row; INDEX
reads it.

INDEX(array, row_num)

MATCH(lookup_value, lookup_array, 0)

fx
=INDEX(D2:D6,MATCH("Fagge

PHC",A2:A6,0))

RETURNS 140

TIP Always end MATCH with 0 for an exact

match. Leave it out and you can get the

wrong row with no error.

12 S U M M A R I S E

MAXIFS
The largest number among the rows

that meet your conditions. MINIFS finds
the smallest.

MAXIFS(max_range, criteria_range1,

criteria1, ...)

fx =MAXIFS(D:D,B:B,"Kano")

RETURNS 140

TIP If no row meets the conditions it returns

0, not an error, so a 0 can mean "nothing

found".

13 S U M M A R I S E

COUNTA
Counts the cells that are not empty,

whatever they hold. Useful for
checking how complete a report is.

COUNTA(value1, [value2], ...)

fx =COUNTA(A2:A6)

RETURNS 5 facilities listed

TIP COUNT counts numbers only. COUNTA

counts text too, and COUNTBLANK counts

the gaps.

14 T I D Y

ROUND
Rounds a number to the number of

decimal places you choose.

ROUND(number, num_digits)

fx =ROUND(E3/D3*100,1)

RETURNS 54.7 positivity, %

TIP Round the final figure, not the steps.

Rounding along the way makes totals drift.

15 D E C I D E

IFS
Checks conditions in order and returns

the result for the first one that is true.

IFS(test1, value1, test2, value2, ...)

fx
=IFS(D3>=100,"High",D3>=50,

"Medium",TRUE,"Low")

RETURNS Medium

TIP End with TRUE,"Low" as a catch-all.

Without it, a value that fits nothing gives

#N/A.

16 L I S T S

SORTBY
Sorts a list by another column, leaving
the original data as it is.

SORTBY(array, by_array1,

[sort_order1], ...)

fx =SORTBY(A2:A6,D2:D6,-1)

RETURNS Fagge PHC

Kubwa PHC

Ikeja GH

and 2 more
below

TIP -1 puts the largest first, 1 the smallest.

SORT sorts by one of the list's own columns.

1 7 T I D Y

TEXTJOIN
Joins values into one cell, with a
separator you choose.

TEXTJOIN(delimiter, ignore_empty,

text1, ...)

fx
=TEXTJOIN(", ",TRUE,

UNIQUE(B2:B6))

RETURNS FCT, Kano, Lagos

TIP TRUE skips blank cells, so you don't get

", ," where data is missing.

18 T I D Y

RIGHT LEFT MID

Takes characters from the end of a
text. LEFT and MID work the same way.

RIGHT(text, num_chars)

fx =RIGHT(A2,3)

RETURNS PHC the facility type

TIP The result is text, even when it looks like

a number. Wrap it in VALUE() to calculate.

19 T I D Y

SUBSTITUTE
Replaces every copy of one piece of
text with another, or with nothing.

SUBSTITUTE(text, old_text, new_text,

[instance])

fx =SUBSTITUTE(A9,"-","")

where A9 holds 0803-123-4567

RETURNS 08031234567

TIP It matches capitals exactly: "fct" will not

replace "FCT".

20 D A T E S

EOMONTH
The last day of the month, a set
number of months away.

EOMONTH(start_date, months)

fx =EOMONTH(C2,0)

RETURNS 31/03/2026

TIP Shows 46112 until you format the cell as

a date. Use 1 for the end of next month.

